

```

10 '
20 'MAN O'WAR by D.MUIR '84
30 '
40 GOTO 1080
50 '
60 'update scoreboards
70 '
80 CLS #1:IF chits>2 THEN chits=0:sh=sh+1:sc=sc+100:cff=-1:IF pl<
3 THEN IF sh>0 THEN IF sh/5=INT(sh/5) THEN pl=pl+1
90 PRINT #1,chits;:SOUND 7,1000,20,5,2,2,7:RETURN
100 CLS #2:PRINT #2,sh;:RETURN
110 CLS #3:PRINT #3,sc;:RETURN
120 CLS #4:IF phits>2 THEN phits=0:pl=pl-1:pff=-1:GOSUB 140
130 PRINT #4,phits;:RETURN
140 CLS #5:PRINT #5,LEFT$(tpl,pl+1);:IF pl=0 THEN lff=-1
150 RETURN
160 '
170 ' move player shell/computer shell
180 '
190 DI:IF NOT pk THEN 240
200 LOCATE pfx,pfy:PRINT " ";
210 IF pfy<5 OR (pfy=14 AND Fnp>3 AND Fnp<8) THEN pk=0:GOTO 240
220 pfy=pfy-1:LOCATE pfx,pfy:PEN 1:PRINT CHR$(145);
230 IF ABS(cx+2-pfx)<2 THEN IF pfy=7 THEN LOCATE pfx,pfy:PEN 2:PR
INT CHR$(238);:chits=chits+1:sc=sc+10:pk=0:GOSUB 80:GOSUB 110
240 IF NOT ck THEN RETURN
250 LOCATE cfx,cfy:PRINT " ";
260 IF cfy>21 THEN ck=0:RETURN
270 cfy=cfy+1:LOCATE cfx,cfy:PEN 1:PRINT CHR$(148);
280 IF ABS(px+2-cfx)<2 THEN IF cfy=21 THEN LOCATE cfx,cfy:PEN 2:P
RINT CHR$(238);:phits=phits+1:ck=0:SOUND 7,1000,20,5,2,2,7:GOSUB
120
290 RETURN
300 '
310 ' move computer boat
320 '
330 DI:cm=cm+1:IF cm=c1 THEN cm=0:cd=-cd:cc=3-cc:c1=10+5*INT(RND*
5):RETURN
340 IF cx+cd>36 OR cx+cd<1 THEN cd=-cd:cc=3-cc
350 cx=cx+cd:LOCATE cx,cy:PRINT USING "&";t(cc);:RETURN
360 '
370 'prime computer gun
380 '
390 DI:IF cff OR pff THEN RETURN
400 IF ck THEN RETURN ELSE ck=-1:cfx=cx+1:cfy=cy+4:IF ABS(px-cx)>
10 OR (Fnc>3 AND Fnc<8) THEN ck=0:RETURN
410 PEN 2:LOCATE cfx,cfy:PRINT CHR$(238);:SOUND 7,2000,10,4,1,1:R
ETURN
420 '
430 ' player move
440 '
450 EVERY 20,2 GOSUB 330
460 EVERY bmi,3 GOSUB 390
470 EVERY 4,0 GOSUB 190
480 IF lff THEN 1010 ELSE IF pff OR cff THEN s=REMAIN(0):s=REMAIN
(2):s=REMAIN(3):bmi=bmi-5:GOTO 560
490 IF INKEY(8)=0 OR INKEY(74)=0 THEN pd=-1:pc=2:GOTO 530
500 IF INKEY(1)=0 OR INKEY(75)=0 THEN pd=1:pc=1:GOTO 530
510 IF INKEY(9)=0 OR INKEY(76)=0 THEN 520 ELSE 530
520 IF pk THEN 530 ELSE pk=-1:pfx=px+1:pfy=py-1:LOCATE pfx,pfy:PE
N 2:SOUND 7,2000,10,4,1,1:PRINT CHR$(238);
530 IF px+pd>36 OR px+pd<1 THEN pd=-pd:pc=3-pc
540 px=px+pd:LOCATE px,py:PRINT USING "&";t(pc);
550 EI:DI:GOTO 480
560 IF cff THEN cff=0:SOUND 7,3000,40,7,0,0,3:LOCATE cx,cy:PRINT
USING "&";t;:FOR j=1 TO 20:NEXT:LOCATE px,py:PRINT USING "&";t;:G
OTO 910
570 IF pff THEN pff=0:SOUND 7,3000,40,7,0,0,3:LOCATE px,py:PRINT
USING "&";t;:FOR j=1 TO 20:NEXT:LOCATE cx,cy:PRINT USING "&";t;:G
OTO 910
580 '
590 ' set up
600 '
610 MODE 1:BORDER 12
620 INK 0,20:INK 1,1:INK 2,26:INK 3,7
630 DEFINT a-r:DEFREAL s:DEFSTR t-z
640 DEF Fnp=VAL(RIGHT$(STR$(pfx),1)):DEF Fnc=VAL(RIGHT$(STR$(cfx)
,1))
650 SYMBOL AFTER 200
660 ENV 1,5,-1,2
670 ENT 1,2,100,2,3,0,2
680 ENV 2,1,5,2,5,0,2,4,-1,2
690 ENT 2,2,100,4,3,0,4
700 RESTORE 950:FOR i=200 TO 223:READ a1,a2,a3,a4,a5,a6,a7,a8:SYM
BOL i,a1,a2,a3,a4,a5,a6,a7,a8:NEXT
710 DIM t(2)
720 t(1)=CHR$(15)+CHR$(2)+CHR$(32)+CHR$(209)+CHR$(210)+CHR$(211)+
CHR$(32)+STRING$(5,8)+CHR$(10)+CHR$(32)+CHR$(212)+CHR$(213)+CHR$(
214)+CHR$(32)+STRING$(5,8)+CHR$(10)+CHR$(15)+CHR$(3)+CHR$(32)+CHR
$(206)+CHR$(207)+CHR$(208)+CHR$(32)
730 t(2)=CHR$(15)+CHR$(2)+CHR$(32)+CHR$(200)+CHR$(201)+CHR$(202)+
CHR$(32)+STRING$(5,8)+CHR$(10)+CHR$(32)+CHR$(203)+CHR$(204)+CHR$(

```

```

205)+CHR$(32)+STRING$(5,8)+CHR$(10)+CHR$(15)+CHR$(3)+CHR$(32)+CHR
$(215)+CHR$(216)+CHR$(217)+CHR$(32)
740 t=SPACE$(5)+STRING$(5,8)+CHR$(10)+SPACE$(5)+STRING$(5,8)+CHR$
(10)+SPACE$(5)
750 u=CHR$(220)+CHR$(221):u1=u+u+u+u+u+u+u+u+u:u1=u1+u1
760 u=CHR$(222)+CHR$(223):u2=u+u+u+u+u+u+u+u+u:u2=u2+u2
770 u=STRING$(40,CHR$(143)):u=u+u
780 u3=SPACE$(4)+CHR$(218)+CHR$(219)+SPACE$(8)+CHR$(218)+CHR$(219
)+SPACE$(4):u3=u3+u3
790 u4=CHR$(218)+CHR$(143)+CHR$(143)+CHR$(219)
800 u5=SPACE$(4)+CHR$(229)+CHR$(229)+SPACE$(8)+CHR$(229)+CHR$(229
)+SPACE$(4):u5=u5+u5
810 u6=CHR$(229)+" "+CHR$(229)
820 tpl=" "+STRING$(3,235):pl=3:sc=0:hits=0:sh=0:bmi=90
830 '
840 ' print screen
850 '
860 LOCATE 1,1:PAPER 2:PEN 3:PRINT u+u1;LOCATE 1,23:PRINT u2+u;
870 PEN 2:PAPER 3:LOCATE 1,1:PRINT "HITS          SHIPS          SCOR
E":LOCATE 1,25:PRINT "HITS          LIVES";
880 WINDOW #1,5,12,1,1:WINDOW #2,18,25,1,1:WINDOW #3,31,40,1,1:WI
NDOW #4,5,14,25,25:WINDOW #5,20,30,25,40:FOR i=1 TO 5:PAPER #i,0:
PEN #i,1:CLS #i:NEXT
890 PAPER 0:PEN 2:LOCATE 1,12:PRINT u3:LOCATE 4,13:PRINT u4:LOCAT
E 14,13:PRINT u4:LOCATE 24,13:PRINT u4:LOCATE 34,13:PRINT u4:PEN
1
900 PRINT CHR$(22)+CHR$(1);:LOCATE 1,12:PRINT u5:LOCATE 4,13:PRIN
T u6:LOCATE 14,13:PRINT u6:LOCATE 24,13:PRINT u6:LOCATE 34,13:PRI
NT u6:PRINT CHR$(22)+CHR$(0);
910 GOSUB 80:GOSUB 100:GOSUB 110:GOSUB 120:GOSUB 140
920 cx=36:cy=5:cd=-1:cc=2:px=1:py=19:pd=1:pc=1:cl=10+5*INT(RND*4)
:cm=0
930 LOCATE cx,cy:PRINT USING "&";t(cc);:LOCATE px,py:PRINT USING
"&";t(pc);
940 GOTO 450
950 DATA 0,0,0,12,8,30,35,43,28,16,124,130,146,186,199,1,0,0,0,48
,32,240,136,168,127,65,73,127,65,73,93,107,17,125,131,1,1,17,57,2
15,252,4,36,252,4,36,116,172,74,255,128,68,96,191,160,223,146,255
,0,146,0,255,0,255,162,255,2,68,8,240,32,224
960 DATA 0,0,0,12,4,15,17,21,56,8,62,65,73,93,227,128,0,0,0,48,16
,120,196,212,63,32,36,63,32,36,46,53,136,190,193,128,128,136,156,
235,254,130,146,254,130,146,186,214,69,255,64,34,16,15,4,7,73,255
,0,73,0,255,0,255,82,255,1,34,6,253,5,251
970 DATA 1,3,7,15,31,63,127,255,128,192,224,240,248,252,254,255,2
55,255,125,73,73,146,0,0,255,255,190,146,146,73,0,0,0,146,73,73
,125,255,255,0,0,73,146,146,190,255,255,68,46,77,85,73,82
980 '
990 'end of game
1000 '
1010 FOR i=1 TO 4:SOUND 7,400+100*i,50,7:FOR j=1 TO 200:NEXT:NEXT
:MODE 0:CLS:PEN 1:PRINT "          FINAL SCORE:":PRINT "          ";sc:PRINT:P
RINT:PRINT "          GALLEONS SUNK:":PRINT "          ";sh:PRINT:PRINT:PRINT "
          ANOTHER GO?"
1020 q$=INKEY$:IF UPPER$(q$)="Y" THEN 1030 ELSE IF UPPER$(q$)<>"N
" THEN 1020 ELSE CLS:END
1030 CLS:LOCATE 1,24:PRINT "INSTRUCTIONS? (Y/N)"
1040 q$=INKEY$:IF UPPER$(q$)="Y" THEN RUN 1180 ELSE IF UPPER$(q$)
<>"N" THEN 1040 ELSE RUN 610
1050 '
1060 'titles
1070 '
1080 MODE 1:INK 0,1:INK 1,22:CLS:PEN 1
1090 DATA 10,10,32,32,32,32,207,223,222,207,32,222,207,223,32,20
7,223,207,32,32,222,207,223,217,32,32,207,32,32,207,32,222,207,22
3,32,207,207,223,32,32,32,32,32,32,32,207,221,220,207,32,207,2
18,207,32,207,207,207,32,32,207,32,207,32,32,32,207,222,223
1100 DATA 207,32,207,218,207,32,207,218,220,32,32,32,32,32,32,
32,207,32,32,207,32,207,32,207,32,207,221,207,32,32,221,207,220,3
2,32,32,207,220,221,207,32,207,32,207,32,207,221,223,-1
1110 RESTORE 1090
1120 READ p:IF p=-1 THEN 1130 ELSE PRINT CHR$(p);:GOTO 1120
1130 LOCATE 5,24:PRINT "INSTRUCTIONS? (Y/N)";
1140 q$=INKEY$:IF UPPER$(q$)="N" THEN 610 ELSE IF UPPER$(q$)<>"Y"
THEN 1140
1150 '
1160 'instructions
1170 '
1180 MODE 1:INK 0,1:INK 1,22:CLS:PEN 1
1190 CLS:PRINT "MAN O'WAR":PRINT "=====":PRINT:PRINT "In this
game, you command a galleon and fight against the computer's shi
ps."
1200 PRINT:PRINT "Your galleon at the bottom of the screen moves a
utomatically back and forth, but you can change its direction usi
ng right and left cursor keys or joystick. Cannonballs are fired u
sing the joystick fire button or copy key."
1210 PRINT:PRINT "Islands may obstruct your line of fire."
1220 PRINT:PRINT "It takes 3 hits to sink a ship and you have 3
ships or lives available. You win an extra life for each 5 ships y
ou sink."
1230 PRINT "You score 10 points for each hit and 100 for sinking a
ship.":PRINT:PRINT "(PRESS ANY KEY)"
1240 q$=INKEY$:IF q$="" THEN 1240 ELSE 610

```